
Highline Excel 2016 Class 05: Introduction To Array Formulas 

 Array 
1) Define Array: Two or more items. 

 Types of Calculations 
1) Aggregate Calculations: 

 Calculations that take two or more items and calculate a single answer such as adding a column 

of numbers. 

2) Array Calculations: 

 Calculations that operate on an array of items rather than single items and which deliver an 

array of answers called a "Resultant Array". 

 Definition of Array Formulas 
1) Formula makes a calculation on an array of items rather than on single items. 

 Example: 

Calculation on single items: B6*B15 

Calculation on an array of items: B4*B13:B16 

 Example: 

1. k argument in LARGE contains a single item: LARGE(A44:A52,2) 

2. k argument in LARGE contains an array of items: LARGE(A44:A52,{1,2,3}) 

2) Calculation results in an array of answer called "Resultant Array". 

 Example: 

B6*B15 evaluates to a single answer: 30422.63 

B10:B14*G10:G14 evaluates to an array of answers: 

{30422.63;12756.17;10110.37;2643.8} 

 Example: 

1.  LARGE(A44:A52,2) evaluates to a single answer: 1102 

2. LARGE(A44:A52,{1,2,3}) evaluates to an array of answers: {1161,1102,1094} 

3) Array Calculations can contain math, comparative, join, or function argument array operations. 

4) "Array Calculations" or "Array Operations”, which deliver an array of items, are distinguished from an 

aggregate calculation, which delivers a single item. 

5) Array Functions are a specific type of array formula that consist of a group of built-in Excel functions (like 

TRANSPOSE) that are programed to make array calculations and deliver an array of answers (Resultant 

Array). 

6) The Resultant Array can be used inside a larger formula, or it can be the final answer. 

7) The final answer from an array formula can either: 

 A single item. 

This happens when the Array Calculation is placed into an aggregate function like with 

the Array Formula: =SUM(LARGE(A44:A52,{1,2,3})). 

 An array of items. 

This happens when the Resultant Array in entered into a range of cells like with the 

Array Formula: =TRANSPOSE(A282:D283). 

8) Some Array Formulas require you to enter the formula with the special keystroke Ctrl + Shift + Enter in 

order for it to evaluate to the correct answer. Next section in this document presents details about 

when Ctrl + Shift + Enter is required. 

  

Resultant 

Array 

 Ctrl + Shift + Enter 
1) Some Array Formulas require you to enter the formula with the special keystroke Ctrl + Shift + Enter in 

order for it to evaluate to the correct answer. 

2) Whether your Array Formula requires Ctrl + Shift + Enter is determined by: 

1. Whether or not the particular function argument that contains the Array Calculation 

requires Ctrl + Shift + Enter. 

Here is a list of function arguments that allow Array Calculations WITHOUT using 

Ctrl + Shift + Enter: 

1. The array_1, array_2, etc. arguments in SUMPRODUCT function 

2. The array argument in AGGREGATE for functions 14 to 19 

3. The lookup_vector argument in the LOOKUP function 

4. The result_vector argument in the LOOKUP function 

5. The array argument in INDEX function 

6. The actual_range argument in CHI.SQ.TEST function 

7. The expected_range argument in CHI.SQ.TEST function 

All other functions that contain Array Calculations require Ctrl + Shift + Enter. 

2. Whether or not the array is an Array Constant. 

If the arrays being used in the Array Calculation are Array Constants, Ctrl + Shift 

+ Enter is NOT required when you are entering the Array Formula into a single 

cell (#3 supersedes if you are entering Array Formula into multiple cells). 

3. Whether or not you are entering the Array Formula into multiple cells simultaneously. 

Any time you enter the Array Formula into multiple cells simultaneously, Ctrl + 

Shift + Enter is required. 

3) How Ctrl + Shift + Enter works: 

 It is usually best to hold down the Ctrl and Shift keys and then hit the Enter key. 

 When you use Ctrl + Shift + Enter, you are telling Excel that the formula is an Array Formula and 

that you need to make an Array Calculation. 

 After you use Ctrl + Shift + Enter, Excel lets you know that it understood you wanted to make an 

Array Calculation by placing curly brackets { } at the beginning and end of the Array Formula. 

You can see the curly brackets in the Formula Bar. You can NOT type them into the formula, 

they only appear after using the keystrokes Ctrl + Shift + Enter. 

 If you forget to use Ctrl + Shift + Enter one of two things will happen: 

1. If the formula is not next to the data set (formula inputs), you will get a #VALUE! error 

2. If the formula is next to the data set (formula inputs), you will get an answer from 

implicit intersection. The answer is almost always wrong. 

 These Built-in Excel function arguments can NEVER perform array calculations: 
1) lookup_value argument in VLOOKUP 

2) lookup_value argument in HLOOKUP 

3) range and sum_range arguments in SUMIF 

4) range argument in COUNTIF 

5) range and average_range arguments in AVERAGEIF 

6) criteria_range and sum_range arguments in SUMIFS 

7) criteria_range argument in COUNTIFS 

8) criteria_range and average_range arguments in AVEARGEIFS 

 Advantages of Array Formulas: Why We Need Array Formulas (How They Are Useful) 
1) We wanted our Excel solution to be more compact. 

 Array Formulas can take calculations that span multiple cells and perform all of the operations in 

a single cell.  

 Example: If you do not need all the individual numbers for a budget and you just need Net 

Income: 

 This is the #1 Reason that we use Array Formula: 

1. To create compact solution to do all the intermediate steps in a single cell. 

 See Array Formula 3 in Video for this example: 

 

 

2) Create Smaller Formulas. 

 Some Array Formulas are much shorter formula than an alternative. 

 See Array Formula 6 in Video for this example: 

 

 

3) Sometimes an Array Formula is the only way to accomplish the goal. 

 Sometimes there is no alternative and you must use an array formula. 

 This example is not in the video. It is an example from a Statistics class where we need to 

calculate Standard Deviation with a single condition. There is no built in function to do this and 

so if you want to do it with a formula you need to use an array formula like this: 

 

 

 

 Drawbacks to Array Formulas 
1) Array Calculations contain many individual calculations (like an entire column of calculations) and when 

the ranges or arrays are very large, calculation time for the formula slows down. If you have an Excel 

Workbook File with many array formulas with large ranges or arrays, the entire file may have slow 

calculation times. 

2) Array Formulas can be very complicated. 

  

Net Income $30,110.36 =SUM(B23:D23)-SUMPRODUCT(ROUND(B13:B16*B23:D23,2))

Totals 9608

=SUMPRODUCT(SUMIFS($B$58:$B$99,$A$58:$A$99,E58:E61))

Totals 9608

=SUMIFS($B$58:$B$99,$A$58:$A$99,E58)+SUMIFS($B$58:$B$99,$A$58:$A$99,E59)+SUMIFS($B$58:$B$99,$A$58:$A$99,E60)+SUMIFS($B$58:$B$99,$A$58:$A$99,E61)

Location

Standard 

Deviation 

Time

Seattle 0.869 {=STDEV.S(IF($A$106:$A$265=D121,$B$106:$B$265))}

Oakland 0.837 {=STDEV.S(IF($A$106:$A$265=D122,$B$106:$B$265))}

SF 1.406 {=STDEV.S(IF($A$106:$A$265=D123,$B$106:$B$265))}

 Types of Array Operators 
1) Math operators: 

 

 Examples of Math Array Operations: 

1. B10:B14*G10:G14 

i. Multiplying each corresponding element in same size ranges. 

2. B10*G10:G14 

i. Multiplying B10 times each cell in the range G10:G14 

3. B10:B14*G10 

i. Multiplying G10 times each cell in the range B10:B14 

2) Comparative Operators 

 

 Examples of Comparative Array Operations: 

1. B10:B14=G10:G14 

2. B10=G10:G14 

3. B10:B14=G10 

3) Join Operator: Ampersand & 

 Examples of Join Array Operations: 

1. B10:B14&G10:G14 

2. B10&G10:G14 

3. B10:B14&G10 

4) Function Argument: 

 When you put more than one item into a function argument that is expecting a single item, you 

are telling the function to deliver multiple answers. For example, if you give the k argument in 

the LARGE function {1,2,3}, it forces LARGE to deliver 3 answers, namely: the first largest, the 

second largest and the third largest. 

 When you put more than one item into a function argument you are making a “Function 

Argument Array Operation”. 

 

Math Operators.

+ Adding.

- Subtracting or Negation.

* Multiplying.

/ Dividing.

^ Raising to an exponent.

( ) Parentheses.

Comparative Operators.

= Equal

<> Not

> Greater than

>= Greater than or equal to

< Less than

<= Less than or equal to

 Size & Shape of Arrays For Array Operation, How Array Operations Evaluate & What Resultant Arrays Look Like 
Full Picture: 

 

Close Up: 

 

Note: If you are multiplying row x row or column x column or table x table (same thing times same thing) the dimensions of the arrays must be the same. 

 

Array Operation The Math That Happens Resultant Array

B2*B5:B8 = 100*0.5 + 100*0.2 + 100*0.1 + 100*0.05 = {50;20;10;5}

B5:B8*B2 = 0.5*100 + 0.2*100 + 0.1*100 + 0.05*100 = {50;20;10;5}

B5:B8*B2:D2 = 100*0.5 + 150*0.5 + 200*0.5 + 100*0.2 + 150*0.2 + 200*0.2 + 100*0.1 + 150*0.1 + 200*0.1 + 100*0.05 + 150*0.05 + 200*0.05 = {50,75,100;20,30,40;10,15,20;5,7.5,10}

B2:D2*B5:B8 = 0.5*100 + 0.5*150 + 0.5*200 + 0.2*100 + 0.2*150 + 0.2*200 + 0.1*100 + 0.1*150 + 0.1*200 + 0.05*100 + 0.05*150 + 0.05*200 = {50,75,100;20,30,40;10,15,20;5,7.5,10}

B5:B8*D5:D8 = 0.5*100 + 0.2*100 + 0.1*100 + 0.05*100 = {50;20;10;5}

A21:D21*A23:D23 = 0.5*100 + 0.2*100 + 0.1*100 + 0.05*100 = {50,20,10,5}

B5:B7*D5:D8 = Error = {50;20;10;#N/A}

A21:C21*A23:D23 = Error = {50,20,10,#N/A}

 Array Constants 
1) Array constants are formula elements that allow you to “hard code” an array of items into a formula. 

1. Example of Array Constants: 

{1,2,3} 

{0,"Below Par";3000,"Par";6000,"Excellent"} 

2. Syntax for Array Constants: 

Curly Brackets house the array:  {   } 

Comma means column   , 

Semi-colon means row    : 

3. Examples of Array Constants in Formulas: 

=SUM(LARGE(B10:B14,{1,2,3})) 

1. This is an Array Formula because the Array Constant {1,2,3} in the k 

argument of LARGE instructs LARGE to deliver 3 numbers. 

2. This Array Formula does not need Ctrl + Shift + Enter because there is 

no Array Operation that generates the array, it is a hard coded array 

(Array Constant). 

=VLOOKUP(C51,{0,"Below Par";3000,"Par";6000,"Excellent"},2) 

3. This is not an Array Formula because the Array Constant represents a 

“lookup table” in the table_array argument in VLOOKUP. Different from 

the Array Constant {1,2,3} in the k argument of LARGE, which instructs 

LARGE to deliver 3 numbers, the Array Constant {0,"Below 

Par";3000,"Par";6000,"Excellent"} does not instruct VLOOKUP to deliver 

multiple items, but rather it is just the table that VLOOKUP chooses a 

single value from. 

4. If the arrays being used in the Array Calculation are Array Constants, Ctrl + Shift + Enter 

is NOT required when you are entering the Array Formula into a single cell. 

i. Examples: 

1. Ctrl + Shift + Enter NOT Required: 

=SUM(LARGE(B10:B14,{1,2,3})) 

or 

=SUM({1,2,3}*{4;4;4}) 

2. This formula requires Ctrl + Shift + Enter because you are entering the 

Array Function into multiple cells simultaneously: 

=FREQUENCY(A311:A323,{250;500;750;1000}) 

 Array Functions 
1) Array Functions are a specific type of array formula that consist of a group of built-in Excel functions that 

are programed to make array calculations and deliver a resultant array of values that can be entered 

into a range of cells using the keystrokes Ctrl + Shift + Enter. 

2) Array Functions in Excel: 

 TRANSPOSE 

Converts a vertical array or range into a horizontal array or range or vice versa. Works 

on one-way or two-way arrays or ranges. 

 MODE.MULT (new in Excel 2010) 

Calculates mode (statistics). Finds the number that occurs most frequently when there 

are multiple such values (multiple modes). 

 FREQUENCY 

Counts how many values are in each category, given the upper values for each category. 

 TREND 

Using the least-squares method for best-fitting data to a straight line, returns an array of 

y values, given these formula inputs: known y values, known x values, and an array of x 

values used to estimate the array of y values. 

 LINEST 

A function that simultaneously returns multiple statistics for single or multiple 

regression, using the least-squares method for best-fitting data to a straight line. 

 MMULT  

Returns the matrix product of two arrays. 

 MUNIT (new in Excel 2013) 

Returns the unit matrix, given a single number. 

 MINVERSE 

Returns the inverse matrix, given a matrix. 

3) For an Array Function to show the correct number of answers, you must anticipate the size of the 

Resultant Array and highlight a range of cells that is the same size as the Resultant Array BEFORE you 

type out the Array Function formula. 

4) After you enter an Array Function with Ctrl + Shift + Enter, you cannot delete the contents of just one 

cell. If you want to delete the contents of a cell, you must delete all the contents of all the cells in the 

array. 

5) Note: Although you must use the keystrokes Ctrl + Shift + Enter to enter an Array Function into a range 

of cells, if you nest the Array Function inside an aggregate function (like SUM, COUNT, MAX, or others) 

you do not have to use the keystrokes Ctrl + Shift + Enter. The one exception seems to be if you put 

TRANSPOSE into SUMPRODUCT. 

6) Example of TRANSPOSE: 

 

We need to go from this: 

 
 

To this: 

 
  

 Steps: 

1. Count Rows and Columns = 2 x 4 

2. Highlight range that is 4 x 2 (rows and columns) 

3. Use Array Function TRANSPOSE and highlight range A282:D283 to create formula: 

=TRANSPOSE(A282:D283) 

4. Must use Ctrl + Shift + Enter because we are entering an array into multiple cells 

simultaneously. 

 Picture: 

 

7) Example of FREQUENCY: 

 

 
 

COGS Operating Administrative Other

46.46% 19.48% 15.44% 4.04%

COGS 46.46%

Operating 19.48%

Administrative 15.44%

Other 4.04%

8) Notes for FREQUENCY (not required for this class, but is required for BI 348 class): 

Here’s what you need to know about how the FREQUENCY array function works: 

 FREQUENCY counts how many numbers are in each category. 

 The bins_array argument contains the upper values for the categories—numbers 

only. 

 The data_array argument contains the values to count—numbers only. 

 Keep in mind the following about categories: 

1) Categories are automatically created. There is no visual indication of 
how the categories are organized. 

2) The first category counts all the values less than or equal to the 
first upper limit. 

3) The middle categories count between a lower limit and an upper limit. 
The lower limit is not included in the category. The upper limit is 

included in the category. 

4) The last category catches all the values that are greater than the 
last upper limit. 

5) There is always one more category than there are bins. 

 FREQUENCY delivers a vertical array. If you need a horizontal array, use the 

TRANSPOSE function to convert a vertical array to a horizontal array. 

 Must use Ctrl + Shift + Enter 

 FREQUENCY ignores empty cells and text. 

 If there are duplicate bins, the duplicates get a count of zero. (You use 

this aspect utilized when you create formulas for unique counting.) 

  



Page 11 of 13 

 Cliff Notes for Array Formulas 
1) Calculation on an array of items rather than single items called an "Array Calculation" or "Array 

Operation". 

2) Calculation results in an array of answer called "Resultant Array". 

3) The Resultant Array can be used inside a larger formula, or it can be the final answer. 

4) Array Formulas require the special keystroke Ctrl + Shift + Enter, except these situations: 

1. One of these five functions: 

i. SUMPRODUCT 

ii. AGGREGATE 

iii. LOOKUP 

iv. INDEX 

v. CHI.SQ.TEST 

2. The arrays are Array Constants 

5) How Ctrl + Shift + Enter works: 

 When you use Ctrl + Shift + Enter, you are telling Excel that the formula is an Array Formula. 

 After you use Ctrl + Shift + Enter, Excel puts curly brackets { } at the beginning and end of the 

Array Formula to indicate that it understood that you were entering an Array Formula. 

 If you DON'T use Ctrl + Shift + Enter and you are required to, you will get a #VALUE! Error or the 

wrong answer. 

6) Specific Array Functions like TRANSPOSE deliver an array of values and require that you highlight a range 

of cells and then use the key stroke Ctrl + Shift + Enter. 

7) Array Formulas can be useful because they can allow: 1) compact Excel solutions, shorter formulas and 

sometimes array formulas are the only way to accomplish the goal. 

8) Drawbacks for Array Formulas: with large Array Calculations, calculation time for the Excel file can slow 

down. 

 Cumulative List of Keyboards Throughout Class: 
1) Esc Key: 

i. Closes Backstage View (like Print Preview). 

ii. Closes most dialog boxes. 

iii. If you are in Edit mode in a Cell, Esc will revert back to what you had in the cell before you put the Cell in 

Edit mode. 

2) F2 Key = Puts formula in Edit Mode and shows the rainbow colored Range Finder. 

3) SUM Function: Alt + =  

4) Ctrl + Shift + Arrow = Highlight column (Current Region). 

5) Ctrl + Backspace = Jumps back to Active Cell 

6) Ctrl + Z = Undo. 

7) Ctrl + Y = Undo the Undo. 

8) Ctrl + C = Copy. 

9) Ctrl + X = Cut. 

10) Ctrl + V = Paste. 

11) Ctrl + PageDown =expose next sheet to right. 

12) Ctrl + PageUp =expose next sheet to left. 

13) Ctrl + 1 = Format Cells dialog box, or in a chart it opens Format Chart Element Task Pane. 

14) Ctrl + Arrow: jumps to the bottom of the "Current Region", which means it jumps to the last cell that has data, 

right before the first empty cell. 

15) Ctrl + Home = Go to Cell A1. 

16) Ctrl + End = Go to last cell used. 

17) Alt keyboards are keys that you hit in succession. Alt keyboards are keyboards you can teach yourself by hitting 

the Alt key and looking at the screen tips. 

i. Create PivotTable dialog box: Alt, N, V 

ii. Page Setup dialog box: Alt, P, S, P 

iii. Keyboard to open Sort dialog box: Alt, D, S 

18) ENTER = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell DOWN. 

19) CTRL + ENTER = When you are in Edit Mode in a Cell, it will put thing in cell and keep cell selected. 

20) TAB = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell RIGHT. 

21) SHIFT + ENTER = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell UP. 

22) SHIFT + TAB = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell LEFT. 

23) Ctrl + T = Create Excel Table (with dynamic ranges) from a Proper Data Set. 

i. Keyboard to name Excel Table: Alt, J, T, A 

ii. Tab = Enter Raw Data into an Excel Table. 

24) Ctrl + Shift + ~ ( ` ) = General Number Formatting Keyboard. 

25) Ctrl + ; = Keyboard for hardcoding today's date. 

26) Ctrl + Shift + ; = Keyboard for hardcoding current time. 

27) Arrow Key = If you are making a formula, Arrow key will “hunt” for Cell Reference. 

28) Ctrl + B = Bold the Font 

29) Ctrl + * (on Number Pad) or Ctrl + Shift + 8 = Highlight Current Table. 

30) Alt + Enter = Add Manual Line Break (Word Wrap) 

31) Ctrl + P =  Print dialog Backstage View and Print Preview 

32) F4 Key = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, 

F4 key will toggle through the different Cell References: 

i. A1 = Relative 

ii. $A$1 = Absolute or “Locked” 

iii. A$1 = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down 

the rows) 

iv. $A1 = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the 

columns) 

33) Ctrl + Shift + 4 = Apply Currency Number Formatting 

34) Tab key = When you are selecting a Function from the Function Drop-down list, you can select the function that 

is highlighted in blue by using the Tab key. 

35) F9 Key = To evaluate just a single part of formula while you are in edit mode, highlight part of formula and hit 

the F9 key. 

i. If you are creating an Array Constant in your formula: Hit F9. 

ii. If you are evaluating the formula element just to see what that part of the formula looks like, 

REMEMBER: to Undo with Ctrl + Z. 

36) Alt, E, A, A = Clear All (Content and Formatting) 

37) Evaluate Formula One Step at a Time Keyboard: Alt, M, V 

38) Keyboard to open Sort dialog box: Alt, D, S 

39) Ctrl + Shift + L = Filter (or Alt, D, F, F) = Toggle key for Filter Drop-down Arrows 
40) Ctrl + N = Open New File 

41) F12 = Save As (Change File Name, Location, File Type) 

42) Import Excel Table into Power Query Editor: Alt, A, P, T 

43) Ctrl + 1 (When Chart element in selected): Open Task Pane for Chart Element 

44) F4 Key = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, 

F4 key will toggle through the different Cell References: 

i. A1 = Relative 

ii. $A$1 = Absolute or “Locked” 

iii. A$1 = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down 

the rows) 

iv. $A1 = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the 

columns) 

45) Keyboard to open Scenario Manager = Alt, T, E 

46) Ctrl + Tab = Toggle between Excel Workbook File Windows 

47) Ctrl + Shift + F3 = Create Names From Selection 

48) Ctrl + F3 = open Name Manager 

49) F3 = Paste Name or List of Names 

50) Alt + F4 = Close Active Window 

51) Window Key + Up Arrow = Maximize Active Window 

 

New In This Video: 

52) Ctrl + Shift + Enter = Keystroke to enter Array Formulas that: 1) have a function argument that requires it, or 2) 

whether or not you are entering the Resultant Array into multiple cells simultaneously. 

53) Ctrl + / = Highlight current Array 


